

永田 靖 著

「サンプルサイズの決め方」

— 補助資料 —

Excel による検出力とサンプルサイズの計算

芳賀 敏郎

目 次

0	まえがき	1
3	1 つの母平均の検定 — 母分散が既知の場合	2
3.1	考え方	2
3.2	検出力の計算方法	3
3.3	サンプルサイズの設計方法	4
(1)	検出力の計算表を用いる方法	4
(2)	計算式を用いる方法	5
4	1 つの母分散の検定	6
4.1	考え方	6
4.2	検出力の計算方法	7
4.3	サンプルサイズの設計方法	7
(1)	検出力の計算表を用いる方法	7
(2)	近似計算式を用いる方法	8
5	1 つの母平均の検定 — 母分散が未知の場合	9
5.1	考え方	9
5.2	検出力の計算方法	11
5.3	サンプルサイズの設計方法	11

6	2つの母分散の比の検定	12
6.1	考え方	12
6.2	検出力の計算方法	14
6.3	サンプルサイズの設計方法	14
7	2つの平均値の差の検定	15
7.2	検出力の計算方法	15
7.3	サンプルサイズの設計方法	15
8	対応がある場合の母平均の差の検定	16
9	1元配置分散分析 — 誤差分散が既知の場合	16
9.2	検出力の計算方法	16
9.3	サンプルサイズの設計方法	18
10	1元配置分散分析 — 誤差分散が未知の場合	18
10.2	検出力の計算方法	18
10.3	サンプルサイズの設計方法	19
11	その他の手法	19
11.1	母不良率の検定	19
(1)	考え方	19
(2)	検出力の計算方法	21
(3)	サンプルサイズの設計方法	22
11.2	回帰係数の検定	23
(1)	検出力の計算方法	23
(2)	x の平方和 S_{xx} の設計方法	24
(3)	補足	25
12	区間推定にもとづくサンプルサイズの設計方法	25
12.1	1つの母平均の区間推定に基づくサンプルサイズの設計方法 (母分散既知)	25
12.2	1つの母分散の区間推定に基づくサンプルサイズの設計方法	26
12.3	1つの母平均の区間推定に基づくサンプルサイズの設計方法 (母分散未知)	27
12.4	2つの母分散の比の区間推定に基づくサンプルサイズの設計方法	28
12.5	2つの母平均の差の区間推定に基づくサンプルサイズの設計方法	29
12.6	対応のある場合の母平均の差の区間推定に基づくサンプルサイズの設計方法	30
13	補遺	31
13.1	統計数値表	31
13.2	検出力の計算表の作成	32
13.3	スクロールバー	34
13.4	ユーザー定義関数	35
13.5	JMP による検出力の計算	36

0 まえがき

得られたデータについての解析方法を解説する本は星の数ほどあるが、どのようにデータを取ったら良いか、また、サンプルサイズの決め方を解説した本は少ない。

最近出版された

永田靖、「サンプルサイズの決め方」、朝倉書店(2003)

は、このテーマについて体系的に説明した貴重な本である。単に手法だけでなく、基本となる考え方や、数理統計学的な基本を懇切丁寧に説明されている。

そこでは説明されている計算手順を理解するために、読者は統計数値表と電卓で数値的に確認するようになっている。

Excel を利用する読者のために作成したのがこの補助テキストである。

原著では、統計数値表と電卓で計算できるように、主として近似式が用いられている。

この補助資料では、非心 t 分布、非心 χ^2 分布、非心 F 分布の確率を計算する VBA マクロを準備したので、すべて正確な計算値が得られる計算手順を説明し、条件を入力するだけで検出力が求められる計算表を提供する。

また、サンプルサイズの計算には、検出力の計算表で必要な検出力が得られる n を試行錯誤で探索するという単純な方法を用いた。試行錯誤を簡単に実行するために「スクロールバー」を採用した。

原著の近似方法は第3,4章だけについて計算表を示した。

このテキストの章、式と例の番号は原著の番号に合わせてある。ただし、各章の最初の節は、原著とは別のアプローチで、その章で扱う問題について説明することとした。

また、このテキストの図と表は原著と区別するために 表示 と呼ぶことにする。

最後の章に、原著とは直接関係の無い事項を補遺として加えた。

ここに示す Excel の計算シートを利用する前に、原著を精読して計算の根拠を十分に理解して欲しい。機械的に数値を入力して結果を利用することは、誤用になる危険がある。

Excel 計算シートと VBA によるマクロの作成には十分に注意を払ったが、誤りが皆無とは言えない。また、不自然な入力（たとえば n が負）に対する十分は省略した。

不具合を発見された方、改善を考案された方は、下記にご連絡下さい。

SGR00643@nifty.com

3 1つの母平均の検定 — 母分散が既知の場合

3.1 考え方


個々の観測値 x が平均値 μ , 標準偏差 σ の正規分布 $N(\mu, \sigma^2)$ に従うものとする . この章では , σ が既知の場合を考える .

原著では , 対立仮説が両側 , 上側 , 下側の順に説明しているが , ここでは , まず対立仮説が上側の場合を詳しく説明する . 下側の場合は符号を逆にするだけであるから , 計算表を示すだけで , 説明は省略する . 最後に両側の場合を説明する .

図3.1(p.33), 図3.2(p.34) には2つの母集団の母平均の違いにより , 母集団0 が $m = (\mu_0 + \mu)/2$ 以下になる確率と , 母集団0 が μ 以下になる確率が示されている .

表示3.1は , Δ , n , α (片側) を入力すると , 2つの母集団の分布と平均値の分布を描き , 図3.1, 3.2 のハッチ部分の確率を求めると共に , 検出力を求めるものである .

表示3.1: 母集団と平均値の分布と検出力の関係を表わすグラフ



左上が帰無仮説 μ_0 に従う x の分布で , 右上は対立仮説 μ の分布である . ただし , 標準偏差は $\sigma = 1$ とする .

Δ に 1.684 または 1.282 を入力すると , 上半分が 原著の図3.1 または 図3.2 となり , 右上の数値が 0.80 または 0.90 となる .

n 個の平均 \bar{x} の標準偏差は σ/\sqrt{n} となり , その分布は表示3.1 の下のグラフのようになる .

$\alpha = 0.05$ を入力すると、標準正規分布の上側 5%点 1.645 が計算される。帰無仮説の平均の分布で $1.645/\sqrt{n} = 0.548$ の位置に縦の線が引かれている。帰無仮説が正しいとき、この縦線よりも上側の確率が第1種の誤りの確率 $\alpha = 0.05$ である。

右の対立仮説の平均の分布で、この縦線よりも上側の確率が検出力 $1 - \beta$ 、下側の確率が第2種の確率 β である。

Excel ファイルでは、 Δ , n , α を変更すると、自動的に検出力の値と下のグラフが変更されるようになっている。

3つのパラメータを色々と変化すると、

Δ が増えたと対立仮説の分布が右に移動し、検出力が上がる、

n が増えたと平均値の分布が狭くなり、検出力が上がる。

α を小さくすると、有意となる境界線が右に移動し、検出力が下がる。

などが分かるであろう。最後は、第1種の誤差の確率 α を小さくすると、第2種の誤差の確率 β が大きくなることを意味している。

3.2 検出力の計算方法

式(3.9), (3.11), (3.6), には3通りの対立仮説に対して、 n, α, Δ_0 から検出力 $1 - \beta$ を計算する式が示されている。

これらの計算をする Excel シートを表示3.2 に示す。

表示3.2: 検出力の計算表

	A	B	C	D	E	F	G	H
2				1 - β				
3	n	α	Δ	上側	下側	両側		
4	9	0.05	0.6	0.562	0.000	0.437		例3.5(p.36)
5	9	0.05	-0.6	0.000	0.562	0.437		例3.7(p.38)
6	9	0.05	0.6	0.562	0.000	0.437		例3.3(p.34)
7								
8	D4	=1-NORMSDIST(NORMSINV(1-B4)-C4*SQRT(A4))						
9	E4	=NORMSDIST(-NORMSINV(1-B4)-C4*SQRT(A4))						
10	F4	=NORMSDIST(-NORMSINV(1-B4/2)-C4*SQRT(A4))						
11		+1-NORMSDIST(NORMSINV(1-B4/2)-C4*SQRT(A4))						

n, α, Δ_0 のセルに値を入力すると、3通りの検出力 $1 - \beta$ が右の3つのセルに求められる。

8 行目以降に、4 行目の検出力を計算する式が示されている。

上側片側検定の検出力を求める式 (3.11)

$$1 - \beta = \Pr(u \leq (-z_\alpha - \sqrt{n}\Delta)) \quad (3.11)$$

を求めているのがセル D4 である。

NORMSDIST(u) は標準正規分布の下側確率を求める関数である。たとえば、NORMSDIST(1.96) は 0.975 となる。

NORMSINV(P) は標準正規分布の下側確率が P となる値を求める関数である。たとえば、NORMSINV(0.025) は -1.96 となる。上側確率 z_α が求めたいときは、NORMSDIST(1-P) としなければならない。

表示 3.1 は、例 3.5 に対応している。

両側検定の場合は、片側の確率を $\alpha/2$ とし、大きい方と小さい方の検出力を計算するために、2 つの成分の合計となる。

表 3.1 ~ 3 と 図 3.3 ~ 5 は Excel ファイルのシート「3 章」に求められている。

表の上の α 、表側の Δ 、表頭の n を変更すると、表の内容とグラフも自動的に更新されるようになっている（ただし、 Δ の変化範囲を変更するときは横軸の目盛りを修正する必要がある）。

これらの計算表の作り方は、§13.2 で説明する。

以下の章に含まれる同様の表とグラフについても同様である。

3.3 サンプルサイズの設計方法

原著に書かれている方法を説明する前に、表示 3.2 の検出力の計算表を使う方法を説明する。

(1) 検出力の計算表を用いる方法

表示 3.1 の計算表で、 n を試行錯誤で変化し、検出力が希望する値を超える最小の n を探索する。

表示3.3は、表示3.2の右に、Excelに標準的に具備されているスクロールバーを設定したものである¹。

表示3.3: 試行錯誤によるサンプルサイズの計算表

	A	B	C	D	E	F	G	H
13				1-β				
14	n	α	Δ	上側	下側	両側		
15	44	0.05	0.5	0.953	0.000	0.913		例3.10(p.40)
16	25	0.05	-0.5	0.000	0.804	0.705		例3.11(p.41)
17	11	0.05	1.0	0.953	0.000	0.913		例3.9(p.40)

表示3.3の α , Δ を入力する．右のスクロールバーの左右の三角マークをクリックすると n が増減し，検出力が変化する．目的に応じた検出力が希望する値を超える n を求める．
 n を大きく変化させたいときは，スクロールバーの中央のバーを左右にドラッグする．
表示3.3 には例3.9, 3.10, 3.11 に対する解が求められている．上側の検出力が 0.95 を，下側の検出力が 0.80 を，両側の検出力が 0.90 を超える n を求めた結果である．

(2) 計算式を用いる方法

原著の式 (3.15), (3.17), (3.19) に従って，サンプルサイズを求める計算表を表示3.4に示す．

表示3.4: サンプルサイズの計算表

	A	B	C	D	E	F	G	H
19				n				
20	α	β	Δ	片側	両側			
21	0.05	0.05	0.5	44	52			例3.10(p.40)
22	0.05	0.2	0.5	25	32			例3.11(p.41)
23	0.05	0.1	1.0	9	11			例3.9(p.40)
24								
25	D15	=ROUNDUP(((NORMSINV(1-A21)+NORMSINV(1-B21))/C21)^2,0)						
26	E15	=ROUNDUP(((NORMSINV(1-A21/2)+NORMSINV(1-B21))/C21)^2,0)						

α , β , Δ_0 を入力すると，両側検定，片側検定に対する n が求められる．原著では片側検定で Δ が+と- の場合が示されているが，表示3.3 では Δ は正の値を取る場合について計

¹ スクロールバーの設定の方法は，§13.3 で説明する．

算する．

ここで， $\text{ROUNDUP}(\quad, 0)$ は，小数点以下の桁数を 0 として，切上げる関数である．


このようにして得られた n に対する検出力は， n を表示 3.2 に入力することに求められる．

4 1つの母分散の検定

4.1 考え方

n, Δ, α (片側) を入力すると，帰無仮説と対立仮説の分布のグラフを描くプログラムを表示 4.1 に示す．

表示 4.1: 3つの χ^2 乗分布と検出力



中央の曲線は帰無仮説 $\sigma_0 = 1.0$ に対応する自由度が $\phi = n - 1$ の χ^2 分布で，上側と下側の確率が，それぞれ， α の位置 (16.919 と 3.323) に縦線が引かれている．

右の曲線は対立仮説 $\sigma = \Delta\sigma_0 = \Delta$ に対応する χ^2 分布である．上側の棄却限界線より大きい部分の面積が検出力で，グラフの上の上側の欄に 0.583 と求められている．

左の曲線は対立仮説 $\sigma = \sigma_0/\Delta = 1/\Delta$ に対応する χ^2 分布である．下側の棄却限界線より小さい部分の面積が検出力で，グラフの上の下側の欄に 0.413 と求められている．

上側の検出力と下側の検出力は等しくないことに注意．

4.2 検出力の計算方法

表示4.2は、式(4.8), (4.11), (4.15) に従って、 n , α , Δ から 検出力 $1 - \beta$ を計算する表である。

表示4.2: 検出力の計算表

	A	B	C	D	E	F	G	H
2				1- β				
3	n	α	Δ	上側	下側	両側		
4	10	0.05	2	0.896	0.000	0.855		例4.3(p.49)
5	10	0.05	0.5	0.000	0.851	0.710		例4.5(p.51)
6	10	0.05	2	0.896	0.000	0.855		例4.1(p.47)
7								
8	D4	=CHIDIST(CHIINV(\$B4,\$A4-1)/\$C4^2,\$A4-1)						
9	E4	=1-CHIDIST(CHIINV(1-\$B4,\$A4-1)/\$C4^2,\$A4-1)						
10	F4	=1-CHIDIST(CHIINV(1-\$B4/2,\$A4-1)/\$C4^2,\$A4-1)						
11		+CHIDIST(CHIINV(\$B4/2,\$A4-1)/\$C4^2,\$A4-2)						

検出力を求める計算式の中の CHIDIST は χ^2 分布の上側確率を求める関数, CHIINV は χ^2 分布の上側100P%点 $\chi^2(\phi, P)$ を求める関数である。

例4.1, 4.3, 4.5 に対する検出力が計算されている。

表示4.2 は、片側検定の場合について計算するものであって、 $\Delta = 2.0$ とすれば、例4.3, 例4.5 に対する解が求められている。

両側検定の例4.1 に対する解は、 $\Delta = 2$, $\alpha = 0.025$ とすれば得られる。

4.3 サンプルサイズの設計方法

前章と同様、まず、表示4.2の検出力の計算表を使う方法を説明する。

(1) 検出力の計算表を用いる方法

表示3.3 と同様の形式の計算表を表示4.3に示す。

使い方は、表示3.4と全く同じである。

上から、上側の検出力を 0.95, 下側の検出力を 0.90, 両側の検出力を 0.90 以上とするために必要な n が求められている。

表示4.3: 試行錯誤によるサンプルサイズの計算表

	A	B	C	D	E	F	G	H
13				1-β				
14	n	α	Δ	上側	下側	両側		
15	9	0.05	2.5	0.963	0.000	0.946	◀ ▶	例4.8(p.54)
16	6	0.05	0.33	0.000	0.933	0.813	◀ ▶	例4.9(p.55)
17	12	0.05	2	0.935	0.000	0.906	◀ ▶	例4.7(p.53)
18	14	0.05	0.5	0.000	0.965	0.906	◀ ▶	

(2) 近似計算式を用いる方法

原著 §4.3 には、 χ^2 分布を正規分布で近似する方法 (Fisher の近似法) が説明されている。式(4.18) - (4.21) を使って、 n を求める計算表を表示4.4 に示す。

表示4.4: サンプルサイズの計算表

	A	B	C	D	E	F	G	H
20	α	β	Δ	n	n'	1-β		
21	H0: $\sigma^2 = \sigma_0^2$, H1: $\sigma^2 > \sigma_0^2$							式(4.20)
22	0.05	0.05	2.5	8.87	8	0.945		例4.8(p.54)
23					9	0.963		
24	H0: $\sigma^2 = \sigma_0^2$, H1: $\sigma^2 < \sigma_0^2$							式(4.21)
25	0.05	0.2	0.33	5.67	5	0.829		例4.9(p.55)
26					6	0.933		
27	H0: $\sigma^2 = \sigma_0^2$, H1: $\sigma^2 \neq \sigma_0^2$							式(4.18, 4.19)
28	0.05	0.1	2.0	11.73	11	0.883		例4.7(p.53)
29					12	0.906		
30	0.05	0.1	0.5	15.03	15	0.931		
31					14	0.906		
32								
33	D15	=0.5*((NORMSINV(1-A15/2)+C15*NORMSINV(1-B15))/(C15-1))^2+1.5						
34	D17	=0.5*((NORMSINV(1-A17/2)+C17*NORMSINV(1-B17))/(C17-1))^2+1.5						
35	D20	=0.5*((NORMSINV(1-A20)+C20*NORMSINV(1-B20))/(C20-1))^2+1.5						
36	D23	=0.5*((NORMSINV(1-A23)+C23*NORMSINV(1-B23))/(C23-1))^2+1.5						

表には、3つの場合がまとめて示されている。目的に応じて適切な行を選び、 α , β , Δ を入力する。

近似値 n と共に、端数を切捨て、切上げた値が n' に求められる。

切捨てた場合の検出力は要求を満たさず、切上げた場合の検出力は要求を満たす場合が多い。このときは、切上げた n' が必要なサンプルサイズとなる。

それに対して、切捨てた場合の検出力も要求を満たす場合は、切捨てた n' より小さい値

を手で入力して、検出力の要求を満たす最小の n を求める (例4.7 の下 参照)² .

5 1つの母平均の検定 — 母分散が未知の場合

5.1 考え方

この章では σ が未知の場合に用いられる t 検定の検出力と必要なサンプルサイズの決め方を取り上げる .

検定には母標準偏差 σ の値を必要としないが、検出力やサンプルサイズの計算には σ の値が必要である .

したがって、過去の解析の結果などからばらつき大きさに対する情報を蓄積して置かねばならない . そのような事前情報が全く無い場合には、比較的小さい n の予備調査 (または実験) で σ の概数を知った上でこの章の手法を適用することになる . また、母集団の違いを確率の違いと考えて必要なパラメータを決めることもできる (原著の例3.1 と例3.2 参照) .

原著では、「検出力の計算には非心 t 分布の分布関数の計算が必要であるが、この厳密な計算は複雑であるから、近似式を使う」としている .

ここでは、非心 t 分布の分布関数を求める ユーザー定義関数 を準備し、それを使って正確な計算をすることにする .

σ が未知の場合の検出力が、 σ 既知の場合の検出力との違いを説明し、最後にグラフで示して非心 t 分布の意味の理解を助けることにする .

σ が既知の場合、帰無仮説と対立仮説で平均値の分布がどのように変るかを表示3.1で示した .

帰無仮説 $H_0 : \mu = \mu_0$ が正しいとき n 個の平均値 \bar{x} と μ_0 との差を平均値の標準誤差 σ/\sqrt{n} で割った値 u_0

$$u_0 = \frac{\bar{x} - \mu_0}{\sigma_0/\sqrt{n}}$$

は標準正規分布に従う .

μ が μ_0 より大きいときは

² 前章の計算は正確な計算であるから、切上げる場合だけを考えれば良かった . この章の方法は近似計算を使っているので、切捨てた値で十分な検出力が得られる場合がある .

$$u_0 = \frac{\bar{x} - \mu_0}{\sigma_0/\sqrt{n}} = \frac{\bar{x} - \mu}{\sigma_0/\sqrt{n}} + \frac{\mu - \mu_0}{\sigma_0/\sqrt{n}} = u + \sqrt{n}\Delta$$

となる． u は標準正規分布に従う．これから， u_0 と u は分布は，平均値が一定値 $\sqrt{n}\Delta$ ずれるだけであることが分かる（表示3.1 参照）．

σ 未知の場合は，

$$t_0 = \frac{\bar{x} - \mu_0}{s/\sqrt{n}} = \frac{\bar{x} - \mu}{s/\sqrt{n}} + \frac{\mu - \mu_0}{s/\sqrt{n}} = t + \sqrt{n}\Delta \frac{\sigma_0}{s}$$

となる． t は自由度 $\phi = n - 1$ の t 分布に従う．ここに， s はサンプルから推定された標準偏差である．


$$s = \sqrt{\frac{S}{n-1}}$$

これを σ 既知の場合と比べると，上の最後の式で第2項の s はデータ毎に変化する値であるから，ばらつきを持つ．したがって， t_0 は t よりもばらつきが大きくなり，それは Δ と共に増加する．

この t_0 の分布が非心 t 分布と呼ばれ， $\sqrt{n}\Delta$ が非心パラメータ λ である．

表示5.1 は，自由度が 10 のとき， t 分布 と，非心パラメータが 1, 2, 3 の非心 t 分布のグラフである．表示3.1 の横軸は $(\mu - \mu_0)/\sigma$ が取られているが，ここでは， $\sqrt{n}\Delta$ とする．

表示5.1: t 分布と非心 t 分布



非心パラメータが大きくなると分布の山は右に移動すると共に，広がりが大きくなり，大きい方に裾を引く非対称性が増えることが分かるであろう．

t 分布の上側確率が 5% の位置に縦線が引かれている．非心 t 分布でこの縦線より右の面積が検出力に対応している．

Excel ファイルでは，自由度と非心パラメータを変更すると分布が再計算され，グラフが自動的に変化するようにになっている．自由度を小さくすると，上に述べた変化が顕著になることが分かる．

5.2 検出力の計算方法

表示 5.2 は n , α , Δ を入力して，3 つの検出力を求めるための計算表である．

表示 5.2: 検出力の計算表

	A	B	C	D	E	F	G	H
2				1 - β				
3	n	α	Δ	上側	下側	両側		
4	9	0.05	0.6	0.500	0.000	0.354		例 5.3 (p. 71)
5	9	0.05	-0.6	0.000	0.500	0.354		例 5.5 (p. 73)
6	9	0.05	0.6	0.500	0.000	0.354		例 5.1 (p. 69)
7								
8	D4	=NTDIST(TINV(B5*2, A5-1), A5-1, C5*SQRT(A5))						
9	E4	=NTDIST(TINV(B5*2, A5-1), A5-1, -C5*SQRT(A5))						
10	F4	=NTDIST(TINV(B5, A5-1), A5-1, C5*SQRT(A5))						
11		+NTDIST(TINV(B5, A5-1), A5-1, -C5*SQRT(A5))						

前にも述べたように，非心 t 分布を用いた正確な値が得られる．

表示 5.3 の下の関数の中の，TINV は，両側確率 P と自由度 ϕ を引数とする関数で， $t(\phi, P)$ が求められる．

また，NTDIST は，非心 t 分布の上側確率を求めるためのユーザー定義関数で，3 つの引数は，t の値，自由度，非心パラメータ $\lambda = \sqrt{n}\Delta$ である³．

5.3 サンプルサイズの設計方法

原著では近似式を使ってサンプルサイズを求めているが，ここでは，表示 5.2 の検出力の計算表で，スクロールバーを使ってサンプルサイズの正確な値を求めることにする．

使い方は表示 3.4 と全く同じである．

³ ユーザー定義関数 NTDIST は VBA で書かれている．その内容は，このテキストの最後の章に示す．

表示5.3: サンプルサイズの計算表

	A	B	C	D	E	F	G	H
13				1-β				
14	n	α	Δ	上側	下側	両側		
15	45	0.05	0.5	0.951	0.000	0.907	◀ ◻ ▶	例5.8(p.76)
16	27	0.05	-0.5	0.000	0.812	0.706	◀ ◻ ▶	例5.9(p.77)
17	13	0.05	1.0	0.960	0.000	0.911	◀ ◻ ▶	例5.7(p.75)

6 2つの母分散の比の検定


6.1 考え方

サンプルの平方和を自由度で割った値 $V = S/\phi$ を 平均平方 と呼ぶことにする（不偏分散または分散と呼ばれることもあるが，母分散 σ^2 と誤解されるおそれがあるので用いないことにする）。

母分散 σ^2 が等しい母集団から取り出した2つのサンプルの平均平方の比はF分布に従う．F分布は分子と分母の平均平方の自由度によって異なる．

表示6.1に3つの自由度の組合せについての分布を示す．


表示6.1: いくつかの自由度対に対するF分布のグラフ



2つの母集団の分散が等しくなく， $\Delta = \sigma_1/\sigma_2$ のとき，平均平方の比を Δ^2 で割るとF分布になる．すなわち，F分布のグラフで，横軸を Δ^2 倍に伸ばし，縦軸を $1/\Delta^2$ に縮小すれば良い．

表示6.2 は、2つのサンプルサイズ n_1, n_2 、母標準偏差の比 $\sigma_1/\sigma_2 = \Delta$ 、検定の α を入力すると、帰無仮説と対立仮説に対する F の分布のグラフが描かれるプログラムの出力である。

表示6.2: 2つの母分散の比の検定のグラフ化



中央の曲線が帰無仮説 $\Delta = 1$ に対する分布である。上側と下側の確率が、それぞれ、 α である値 $F(\phi_1, \phi_2; \alpha)$, $F(\phi_1, \phi_2; 1 - \alpha)$ (棄却限界値) が上に示されている⁴,

グラフには棄却限界値の位置に縦線が引かれている。

右の曲線は Δ 、左の曲線は $1/\Delta$ に対する分布である。棄却限界値の外側の確率が検出力で、その値が上に示されている。

これまでのグラフと同様、Excel ファイルでは、4つのパラメータを変化させたときの変化を見ることができる。

表示6.2は分母と分子の自由度が異なる場合であって、標準偏差が2倍になったときの検出力と、半分になったときの検出力は等しくない。

しかし、分母と分子の自由度が等しい場合は、両側の検出力は等しくなる。

⁴ この確率は FINV(上側確率, 分子の自由度, 分母の自由度) で求められる。下側確率が 0.05 のときは上側確率を 0.95 とすれば良い

6.2 検出力の計算方法

表示6.3 は n_1 , n_2 , α , Δ を入力して, 3つの検出力を求めるための計算表である.

表示6.3: 検出力の計算表

	A	B	C	D	E	F	G	H	I
2					1- β				
3	n_1	n_2	α	Δ	下側	上側	両側		
4	10	8	0.05	2.0	0.558	0.000	0.412		例6.3(p.92)
5	10	8	0.05	0.5	0.001	0.592	0.000		例6.5(p.94)
6	10	8	0.05	2.0	0.558	0.000	0.412		例6.1(p.91)
7									
8	E4	=FDIST(FINV(\$C4,\$A4-1,\$B4-1)/\$D4^2,\$A4-1,\$B4-1)							
9	F4	=1-FDIST(FINV(1-\$C4,\$A4-1,\$B4-1)/\$D4^2,\$A4-1,\$B4-1)							
10	G4	=FDIST(FINV(\$C4/2,\$A4-1,\$B4-1)/\$D4^2,\$A4-1,\$B4-1)							
11		+1-FDIST(FINV(1-\$C4/2,\$A4-1,\$B4-1)/\$D4^2,\$A4-1,\$B4-2)							

ここで用いられている, FINV 関数については前節で説明した.

FDIST は, F の値と自由度を引数とする関数で, 上側確率が求められる.

6.3 サンプルサイズの設計方法

表示6.4 に, これまでと同じ考えでサンプルサイズを求める計算表を示す.

表示6.4: サンプルサイズの計算表

	A	B	C	D	E	F	G	H	I
13					1- β				
14	n_1	n_2	α	Δ	上側	下側	両側		
15	15	15	0.05	2.0	0.808	0.000	0.706	◀▶	例6.8(p.97)
16	11	11	0.05	0.333	0.000	0.952	0.911	◀▶	例6.9(p.98)
17	15	15	0.05	2.5	0.952	0.000	0.911	◀▶	例6.7(p.96)
18	24	24	0.05	0.5	0.000	0.946	0.902	◀▶	

手順は表示4.3 と全く同じであるが, n_2 は n_1 に等しくなるように設定されている.

表示6.4の計算表は $n_1 = n_2$ という制約のもとで検出力が要件を満たす n を求めるものであった.

この制約を除いて, $n_1 + n_2 = \text{一定}$ という条件で, 検出力を最大にする n_1 , n_2 を求めると, 異なる解の得られる場合がある.

例6.8の場合,表示6.5に示すように, $n_1 = n_2 = 14$ の検出力0.781よりも, $n_1 = 13, n_2 = 15$ の検出力0.783がわずかながら大きくなる.この差は実務的には意味がないが, $n_1 = n_2$ が必ずしも最適ではないというのは,面白い.

表示6.5: $n_1 \neq n_2$ の場合

	A	B	C	D	E	F	G	H	I
20					1- β				
21	n1	n2	α	Δ	上側	下側	両側		
22	14	14	0.05	2.0	0.781	0.000	0.671		例6.8(p.97)
23	13	15	0.05	2.0	0.783	0.000	0.678		
24	12	16	0.05	2.0	0.781	0.000	0.679		
25	11	11	0.05	0.3	0.000	0.952	0.000		例6.9(p.98)
26	10	12	0.05	0.3	0.000	0.951	0.000		

例6.9の場合は, $n_1 = n_2 = 11$ が最適である.

7 2つの平均値の差の検定

7.2 検出力の計算方法

§5.2の場合と同様に,非心t分布を使って検出力の正確な値を求める.

表示7.1: 検出力の計算表

	A	B	C	D	E	F	G	H	I
2					1- β				
3	n1	n2	α	Δ	上側	下側	両側		
4	10	8	0.05	0.6	0.332	0.002	0.221		例7.3(p.110)
5	10	8	0.05	-0.6	0.002	0.332	0.221		例7.5(p.112)
6	10	8	0.05	0.6	0.332	0.002	0.221		例7.1(p.108)
7									
8	G4	=NTDIST(TINV(C4*2,A4+B4-2),A4+B4-2,D4*SQRT(A4*B4/(A4+B4)))							
9	F4	=NTDIST(TINV(C4*2,A4+B4-2),A4+B4-2,-D4*SQRT(A4*B4/(A4+B4)))							
10	E4	=NTDIST(TINV(C4,A4+B4-2),A4+B4-2,D4*SQRT(A4*B4/(A4+B4)))							
11		+NTDIST(TINV(C4,A4+B4-2),A4+B4-2,-D4*SQRT(A4*B4/(A4+B5)))							

7.3 サンプルサイズの設計方法

表示7.2は,前章と同様に $n_1 = n_2$ に設定してある.

表示7.2: サンプルサイズの計算表

	A	B	C	D	E	F	G	H	I
13					1-β				
14	n1	n2	α	Δ	上側	下側	両側		
15	11	11	0.05	1.5	0.960	0.000	0.917	◀▶	例7.8(p.115)
16	10	10	0.05	-1.2	0.000	0.825	0.718	◀▶	例7.9(p.115)
17	22	22	0.05	1.0	0.947	0.000	0.900	◀▶	例7.7(p.114)

8 対応がある場合の母平均の差の検定

この場合は、§5 (1つの母平均の検定 — 母分散が未知の場合) と全く同じ計算表が利用できる、省略する。

9 1元配置分散分析 — 誤差分散が既知の場合

9.2 検出力の計算方法

検出力の計算表と例についての計算結果を表示9.1 に示す (7,8 行目については補遺で説明する)。

表示9.1: 検出力の計算表

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
2	μ_1	μ_2	μ_3	μ_4	μ_5	S	α	k	n	σ ²	Δ	λ	1-β		
3	10	11.6	12			2.24	0.05	3	5	1.414	1.12	5.6	0.553		例9.1(p.133)
4															
5					d	S	α	k	n	σ ²	Δ	λ	1-β		
6					0	0.0	0.05	3	5	1.414	0.00	0.0	0.050		例9.2(p.134)
7					2	2.0	0.05	6	5	1.414	1.00	5.0	0.363		補遺
8					2	2.8	0.05	6	5	1.414	1.40	7.0	0.501		
9															
10	F3:	=DEVSQ(A3:E3)								K3:	=F3/J3^2				
11	F6:	=E6^2/2								L3:	=I3*K3				
12	F8:	=H8*(H8+1)/(12*(H8-1))*E8^2								M3:	=Ncdist(CHIINV(G3,H3-1),H3-1,L3)				

例9.1 は各水準の母平均 $\mu_1, \mu_2, \dots, \mu_k$ を与えて、検出力を求める例である。そこでは、対立仮説の μ_i の違いを

$$S = \sum_{i=1}^k (\mu_i - \bar{\mu})^2$$

で定量化する．

F3 の S は , μ_i から平方和を求める関数 DEVSQ で簡単に求めることができる．

水準数 k が5を超えるとときは , D 列と E 列の間に列を挿入する．

例9.2 は各水準の母平均の最大値と最小値の差 d を与えて , 検出力を求める例である．
ここでは , 平方和を最小値とする , $\mu = -d/2, 0, \dots, 0, d/2$ に対する S

$$S = \left(\frac{-d}{2}\right)^2 + 0^2 + \dots + 0^2 + \left(\frac{d}{2}\right)^2 = \frac{1}{2}d^2$$

を F6 に求める．

後半の G 列以降は , 対立仮説の表わし方には無関係である．

α, k, n, σ_0 を入力すると , K 列と L 列に

$$\Delta = \frac{S}{\sigma_0^2}$$

$$\lambda = n\Delta$$

が計算される．M 列の検出力 $1 - \beta$ は , 非心 χ^2 分布の累積確率を求めるユーザー定義関数 NCDIST を使って計算される．

NCDIST 関数の最初のパラメータは $\chi^2(k-1, \alpha)$ である．2 番目と3 番目のパラメータは $k-1, \lambda$ である．

補遺

例9.2 では , k 個の水準の範囲を d で , 最大と最小の水準の効果が $-d/2, d/2$, その他の水準の効果が 0 として $S = d^2/2$ を計算した．

それに対して , $-d/2 \sim d/2$ の間で効果が等間隔になっているとしたい場合は

$$S = \frac{k(k+1)}{12(k-1)}d^2$$

とし , この関数を F 列に入力すれば良い．

この2 つの対立仮説の関係を , $k = 6$ について左のグラフで示す．

						k	3	4	5	6	7	8	9	10
○			◎		○	両端	0.50	0.50	0.50	0.50	0.50	0.50	0.50	0.50
○	○	○	○	○	○	等間隔	0.50	0.56	0.63	0.70	0.78	0.86	0.94	1.02

両端だけに効果のあるときは S を求める式の d^2 の係数は 0.5 で一定であるが、等間隔のときは、右の表のように k によって変化する。

表示 9.1 の 7, 8 行目には、 $k = 6$ のとき、2 つの場合で検出力がどの程度変化するかを示す。

9.3 サンプルサイズの設計方法

考え方はこれまでと全く同じである。

表示 9.2: サンプルサイズの計算表

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
14					d	S	α	k	n	σ_0	Δ	λ	$1 - \beta$		
15					2	2.0	0.05	4	8	1.00	2.00	16.0	0.934	◀ ▶	例 9.4 (p. 138)
16					3	4.5	0.01	5	12	1.73	1.50	18.0	0.838	◀ ▶	例 9.5 (p. 139)
17					3	5.6	0.01	5	9	1.73	1.88	16.9	0.804		補造

例 9.4 と 例 9.5 は検出力を 0.90, 0.80 とするために必要なサンプルサイズを求めるものである。

17 行目には、例 9.5 で $d = 3$ の範囲で水準平均が等間隔である場合に必要なサンプルサイズがどのように変化するかを見たものである。必要なサンプルサイズは 12 から 9 に減少することが分かる。

10 1 元配置分散分析 — 誤差分散が未知の場合

10.2 検出力の計算方法

誤差分散が未知の場合の検出力の計算表を表示 10.1 に示す。

表示 9.1 と比べて違うのは、M 列の検出力を計算する関数だけである。

非心カイ 2 乗分布の代わりに、非心 F 分布の関数 NFDIST を用いる。

最初のパラメータは $F(k - 1, k(n - 1); \alpha)$ を求める FINV 関数となる。 $k(n - 1)$ は 1 元配置分散分析の誤差の自由度である。

また、自由度のパラメータが 2 つになる。

表示 10.1: 検出力の計算表

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
2	μ 1	μ 2	μ 3	μ 4	μ 5	S	α	k	n	σ 0	Δ	λ	$1-\beta$		
3	10	11.6	12			2.24	0.05	3	5	1.414	1.12	5.6	0.447		例 10.1 (p. 151)
4															
5					d	S	α	k	n	σ 0	Δ	λ	$1-\beta$		
6					2.00	2.0	0.05	3	5	1.414	1.00	5.0	0.405		例 10.2 (p. 152)
7															
8	F3:	=DEVSQ(A3:E3)						K3:	=F3/J3^2						
9	F6:	=E6^2/2						L3:	=I3*K3						
10								M3:	=NFDIST(FINV(G3, H3-1, H3*(I3-1)), H3-1, H3*(I3-1), L3)						

10.3 サンプルサイズの設計方法

サンプルサイズの求め方も検出力の計算と同様に、表示 9.2 と同じである。

表示 10.2 は、§9 の「誤差分散が既知の場合」と §10 「誤差分散が未知の場合」の両方に共通に使える計算表を示す。

表示 10.2: サンプルサイズの計算表

	Q	R	S	T	U	V	W	X	Y	Z	AA	AB
2									$1-\beta$			
3	d	S	α	k	n	σ 0	Δ	λ	σ 既知	σ 未知		
4	2	2.0	0.05	3	5	1.41	1.00	5.0	0.504	0.401		例 10.2 (p. 152)
5	2	2.0	0.05	4	9	1.00	2.00	18.0	0.959	0.933		例 10.4 (p. 157)
6	3	4.5	0.01	5	13	1.73	1.50	19.5	0.875	0.822		例 10.5 (p. 139)

この計算表により、誤差分散が既知と未知で検出力がどの位異なるかを知ることができる。

11 その他の手法

11.1 母不良率の検定

(1) 考え方

母不良率が P の母集団から n 個の製品をランダムに選び、その中に含まれる不良品の個数を x とする。

x が二項分布に従うとき、 x である確率は

$$= \text{BINOMDIST}(x, n, P, \text{FALSE})$$

で, x 以下である累積確率は

=BINOMDIST(x , n , P , TRUE)

で, x 以上である累積確率は

=1-BINOMDIST($x-1$, n , P , TRUE)

で求められる.

表示 11.1 の Q, R の列に, $P_0 = 0.4$, $n = 10$ の場合の下側累積確率と上側累積確率を示す.

表示 11.1: 2 項分布の計算表

	M	N	O	P	Q	R	S
2					p0		P
3	p0	0.4		x	下側累積確率	上側累積確率	上側累積確率
4	p	0.6		0	0.0060	1.0000	1.0000
5	n	10		1	0.0464	0.9940	0.9999
6	α	0.05		2	0.1673	0.9536	0.9983
7				3	0.3823	0.8327	0.9877
8	下限	1		4	0.6331	0.6177	0.9452
9	上限	8		5	0.8338	0.3669	0.8338
10				6	0.9452	0.1662	0.6331
11	N7:	=Bin_L(N4, N3, N5)		7	0.9877	0.0548	0.3823
12	N8:	=Bin_U(N4, N3, N5)		8	0.9983	0.0123	0.1673
13	Q3:	=BINOMDIST(P3, \$N\$4, \$N\$3, TRUE)		9	0.9999	0.0017	0.0464
14	R4:	=1-BINOMDIST(P4-1, \$N\$4, \$N\$3, TRUE)		10	1.0000	0.0001	0.0060

帰無仮説 $H_0: P = 0.4$ を片側危険率 α で検定する (片側検定).

累積確率が α より小さいとき太字で表わしている. 実測された x が太字の領域 $x \leq 1$, $8 \leq x$ のとき, 帰無仮説が棄却される. すなわちこの領域が棄却域となる.

対立仮説 H_1 の下で $P = 0.6$ のときの上側累積確率を表示 11.1 の一番右の列に示す. このとき, $x \geq 8$ となる確率は 0.1673 で, これが検出力となる. $x \leq 1$ のときも有意となるが, この確率は $1 - 0.9983 = 0.0017$ で極めて小さい.

上側または下側の累積確率が α 以下になる限界値は $x = 1$, または 8 である. この限界値を表示 11.1 の右半分のような表を作ることなく求めるためのユーザー定義関数を準備した.

ユーザー定義関数 Bin_L, Bin_U は, n , P , α (片側) を引数として, 下側および上側の限界値が求めるものである. 表示 11.1 の N8:N9 はこれらの関数を使って求めたものである.

(2) 検出力の計算方法

原著では二項分布を正規分布で近似して検出力を求めているが、ここではまず、前項で説明した考えに基づいて、計算する。

表示11.2に示すように、左に n , α , P_0 , P を入力すると、3通りの検出力が得られる。

表示 11.2: 検出力の計算表

	A	B	C	D	E	F	G	H	I	J
2					1 - β					
3	n	α	P_0	P	上側	下側	両側			
4	200	0.05	0.1	0.05	0.000	0.796	0.700		例11.1, p. 167	
5	200	0.05	0.1	0.05	0.000	0.837	0.708			0.708
6										
7	E4:	=IF(BINOMDIST(\$A4,\$A4,\$C4,FALSE)>\$B4,0,								
8		1-BINOMDIST(Bin_U(\$A4,\$C4,\$B4),\$A4,\$D4,TRUE))								
9	F4:	=IF(BINOMDIST(0,\$A4,\$C4,FALSE)>\$B4,0,								
10		BINOMDIST(Bin_L(\$A4,\$C4,\$B4),\$A4,\$D4,TRUE))								
11	G4:	=IF(BINOMDIST(0,\$A4,\$C4,FALSE)>\$B4/2,0,								
12		BINOMDIST(Bin_L(\$A4,\$C4,\$B4/2),\$A4,\$D4,TRUE))								
13		+IF(BINOMDIST(\$A4,\$A4,\$C4,FALSE)>\$B4/2,0,								
14		1-BINOMDIST(Bin_U(\$A4,\$C4,\$B4/2)-1,\$A4,\$D4,TRUE))								
15										
16	E5:	NORMSDIST(((D5-C5)-NORMSINV(1-B5)								
17		*SQRT(C5*(1-C5)/A5))/SQRT(D5*(1-D5)/A5))								
18	F5:	=1-NORMSDIST(((D5-C5)+NORMSINV(1-B5)								
19		*SQRT(C5*(1-C5)/A5))/SQRT(D5*(1-D5)/A5))								
20	G5:	NORMSDIST(((D5-C5)-NORMSINV(1-B5/2)								
21		*SQRT(C5*(1-C5)/A5))/SQRT(D5*(1-D5)/A5))								
22		+1-NORMSDIST(((D5-C5)+NORMSINV(1-B5/2)								
23		*SQRT(C5*(1-C5)/A5))/SQRT(D5*(1-D5)/A5))								

例11.1 は対立仮説が $H_1: P \neq P_0$ であるから、左の「両側」に検出力が得られる。原著の解は 0.708 で、ここで得られた解との差は実用上問題ない大きさである。

原著では二項分布を正規分布で近似する方法が用いられている。

その方法を上の例で説明する。

帰無仮説の下で、個数 x は $N(nP_0, nP_0(1-P_0)) = N(200 \times 0.1, 200 \times 0.1(1-0.1)) = N(20, 18)$ の正規分布で近似される。両側危険率 0.05 のとき、 $20 - 1.96\sqrt{18} < x < 20 + 1.96\sqrt{18}$ の範囲外が棄却域となる。

対立仮説で $P = 0.05$ のとき、 x は $N(nP, nP(1-P)) = N(200 \times 0.05, 200 \times 0.05(1-0.05)) = N(10, 9.5)$ の正規分布で近似される。これから、 $x < 20 - 1.96\sqrt{18} = 11.68$ となる

確率は，標準正規分布で $u < (11.68 - 10)/\sqrt{9.5} = 0.546$ となる確率 0.708 となる．大きい方で有意になる確率はほとんど0であるから，これが両側検定の検出率となる．

この式で計算する検出力は表示11.2の 5行目に求められ，その計算式が16-23行目に示されている．

(3) サンプルサイズの設計方法


これまでと同様に，スクロールバーを使って要求される検出力を満たす n を求める．

表示 11.3: サンプルサイズの設計方法

	A	B	C	D	E	F	G	H	I	J
26	n	α	p0	p	上側	下側	両側			
27	292	0.05	0.1	0.05	0.000	0.938	0.902		例11.2, p. 168	
28	302	0.05	0.1	0.05	0.000	0.958	0.901			302
29	455	0.05	0.1	0.15	0.923	0.000	0.902			
30	438	0.05	0.1	0.15	0.939	0.000	0.900			438

これまでと異なるのは， n を大きくすると検出力が単調に増加するとは限らない．
 例11.2の前半で， n を270 から 310 まで変化させて検出力を求めると表示11.4のように鋸状に変化する．これから，検出力が 0.90 を超える最小の n は 292 となる．

表示 11.4: サンプルサイズと検出力の関係



鋸の刃の位置は下側で有意となる限界値が1 変化する n に対応している．

ここに示したように二項分布のように離散値をとる分布の場合には，検出力なども不連続となる．

原著に示されている方法は，離散値の分布を連続な正規分布で近似しているので，鋸の間を通る曲線によって検出力やサンプルサイズを求めていることになる．

表示 11.3 の 28,30 行には原著の方法による計算結果を示す．

11.2 回帰係数の検定

(1) 検出力の計算方法

原著の例 11.3 は， $n = 10$ ， $\alpha = 0.05$ ， $S_{xx} = 16$ ， $\Delta = 0.8$ の場合に，両側の対立仮説に対する検出力を求めるものである．

まず，上側の対立仮説の場合の検出力を求める．

検出力は，自由度が $n - 2$ ，非心パラメータが $\lambda = \sqrt{S_{xx}}\Delta = 3.2$ の非心 t 分布が $t(n - 2, 2\alpha) = 1.860$ より大きい確率として求められる．

その結果は，表示 11.5 の片側の欄の検出力 $1 - \beta$ の欄に 0.898 として求められている．

表示 11.5: 検出力の計算表

	A	B	C	D	E	F	G	H	I	J
38					1 - β					
39	n	α	S_{xx}	Δ	片側	両側				
40	10	0.05	16	0.8	0.898	0.800			例 11.3, p. 172	0.799
41										
42	E4:	=NTDIST(TINV(\$B4*2, \$A4-2), \$A4-2, SQRT(\$C4)*\$D4)								
43	F4:	=NTDIST(TINV(\$B4, \$A4-2), \$A4-2, SQRT(\$C4)*\$D4)								
44		+1-NTDIST(-TINV(\$B40, \$A40-2), \$A40-2, SQRT(\$C40)*\$D40)								

両側の対立仮説の場合の検出力は， $t(n - 2, 2\alpha) = 1.860$ の代わりに $t(n - 2, \alpha) = 2.306$ を用い，2.306 以上の確率と -2.306 以下の確率の合計として求められる．その結果が両側の欄の 0.800 となる．

原著では非心 t 分布の確率を近似式を使って求めているが，表示 11.4 は精確な方法で計算している．

(2) x の平方和 S_{xx} の設計方法

表示11.6の48行目のように, n , α , Δ を入力する. S_{xx} には適当な値 (ここでは 10.0) を入力する. 検出力は 0.603 で, 所定の検出力 0.90 よりも小さい.

表示11.6: S_{xx} の計算表

	A	B	C	D	E	F	G	H	I	J
46					1- β					
47	n	α	S_{xx}	Δ	片側	両側				
48	10	0.05	10.0	0.8	0.746	0.603			初期値	
49	10	0.05	21.5	0.8	0.958	0.900			例11.4, p. 173	21.61
50	13	0.05	20.2	0.8	0.957	0.905	4	d	等間隔	
51	9	0.05	24.0	0.8	0.969	0.917	4	d	3点	
52	14	0.05	20.2	0.8	0.959	0.909	1.2	σ_x	変量	

所定の検出力 0.90 が得られる S_{xx} を試行錯誤で探索すれば良い.

これまでは「サンプルサイズ n の設計方法」であって, 解は整数値に限られていたので, スクロールバーを使って探索をした. スクロールバーの基本機能は整数値だけを扱うので, 連続量 S_{xx} の探索には使えない. そこで, ゴールシークを用いることにする.

トップメニューの「ツール」から「ゴールシーク」を選択すると, 表示11.7のゴールシークのパラメータ入力画面が現れる.

表示 11.7: ゴールシーク画面

「目的セル」に両側の検出力のセル F48 を, 目標値に 0.90 を, 変化させるセルに S_{xx} のセル C48 を指定して実行する. その結果を 49行目に示す.

(3) 補足

前項のようにして、希望する検出力が得られる S_{xx} が計算されたが、実際に実験を計画しようとするとき、具体的に x をどのように取るかという問題にぶつかる。

逆に、 x の範囲が d と限られている場合には、その範囲内で等間隔に x を取るとするときに n を何個にするかという設問に対して n を求めるという方法が考えられる。

限られた範囲で S_{xx} を最大にするのは、 $x = \bar{x} \pm d/2$ で $n/2$ 回ずつ測定すれば良い。この方法は直線関係が確実である場合にはもっとも効率の良い方法であるが、この確信がない場合には、直線性が成立するかを検定するために、3水準で実験する必要がある。そのような場合には、 $x = \bar{x} - d/2, \bar{x}, \bar{x} + d/2$ の3水準にほぼ等しい個数を割り上げるのが良いであろう。

ここに挙げた2つの場合に、 S_{xx} は d, n の関数として次のように求められる。

$$\begin{aligned} S_{xx} &= \frac{n(n+1)}{12(n-1)} d^2 && \text{等間隔} \\ &= \frac{nd^2}{6} && \text{3点} \end{aligned}$$

表示11.6の50行目は、 $d = 4$ の範囲内で x を等間隔に取る場合に必要な n を求めた結果である。

また、51行目は、 $d = 4$ の範囲内で x を3段階に変える場合の結果である。 n が3の倍数である場合は、 $-d/2, 0, d/2$ で同数回実験するが、3の倍数でない場合は、中央の x の実験を少なくするのが良いであろう。

12 区間推定にもとづくサンプルサイズの設計方法

12.1 1つの母平均の区間推定に基づくサンプルサイズの設計方法（母分散既知）

母平均を

$$\bar{x} - z_{\alpha/2} \sqrt{\frac{\sigma_0^2}{n}} < \mu < \bar{x} + z_{\alpha/2} \sqrt{\frac{\sigma_0^2}{n}}$$

として区間推定するとき、信頼区間の幅 $2z_{\alpha/2} \sqrt{\sigma_0^2/n}$ が δ 以下になるようにサンプルサイズ n を設計する。

表示 12.1: 1つの母平均の区間推定に基づくサンプルサイズの設計方法（母分散が既知の場合）

n	α	σ	信頼幅			
7	0.05	2	2.96	◀	▶	例12.1, p.183
D3:	=2*NORMSINV(1-B3/2)*C3/SQRT(A3)					

12.2 1つの母分散の区間推定に基づくサンプルサイズの設計方法

母分散を

$$\frac{S}{\chi^2(n-1, \alpha/2)} < \sigma^2 < \frac{S}{\chi^2(n-1, 1-\alpha/2)}$$

として区間推定するとき，信頼上限と信頼下限の比 $\frac{\chi^2(n-1, \alpha/2)}{\chi^2(n-1, 1-\alpha/2)}$ が δ 以下になるようにサンプルサイズ n を設計する．

表示 12.2: 1つの母分散の区間推定に基づくサンプルサイズの設計方法

n	α	信頼限界比			
28	0.05	2.96	◀	▶	例12.2, p.186
D10:	=CHIINV(B10/2, A10-1)/CHIINV(1-B10/2, A10-1)				


なお，この比は n と α だけで決まる．横軸に n を，縦軸に母標準偏差の 信頼上限/信頼下限 の比をとったグラフを表示12.3 に示す．

n が小さい部分は左の目盛り，大きい部分は右の目盛りに対応する．

縦軸は，母分散の信頼限界の比ではなく，その平方根になっていることに注意．

このグラフから，分散（または標準偏差）の推定値の誤差は予想以上に大きいことに気が付かれたであろう．ばらつきの大きさを精度良く推定するためにはかなりのサンプルサイズが必要である．

表示12.3: n と標準偏差の 上限/下限 の比の関係



12.3 1つの母平均の区間推定に基づくサンプルサイズの設計方法(母分散未知)

母平均を

$$\bar{x} - t(n-1, \alpha) \sqrt{\frac{V}{n}} < \mu < \bar{x} + t(n-1, \alpha) \sqrt{\frac{V}{n}}$$

として区間推定するとき, 信頼区間の幅 $2t(n-1, \alpha) \sqrt{\sigma_0^2/n}$ が δ 以下になるようにサンプルサイズ n を設計する.

上の式で, \sqrt{V} が未知であるから, その期待値で置き換える. V の期待値は σ^2 であるが, \sqrt{V} の期待値は

$$E[\sqrt{V}] = c^* \sigma$$

で, 係数の c^* は V の自由度 ϕ によって異なり,

$$c^* = \frac{\sqrt{2}\Gamma(\phi+1)/2}{\sqrt{\phi}\Gamma(\phi/2)} \quad (12.18)$$

で表わされる. 具体的な数値は原著の表12.2(p.189) および, 添付のExcel ファイルに示されている.

Excel で $\Gamma(x)$ の対数は =GAMMALN で求めることができる.

$$2t(n-1, \alpha) \frac{c^* \sigma}{\sqrt{n}} \leq \delta \quad (12.18)$$

表示 12.4: 1 つの母平均の区間推定に基づくサンプルサイズの設計方法 (母分散が未知の場合)

n	α	σ	信頼幅		
9	0.05	2	2.98		例12.3, p.189
D17:	=2*TINV(B17,A17-1)*C17/SQRT(A17) *SQRT(2/(A17-1))*EXP(GAMMALN((A17)/2)-GAMMALN((A17-1)/3))				

となる n を求める計算表を表示 12.4 に示す .

関数の 2 行目は $\times c^*$ に対応している .


12.4 2 つの母分散の比の区間推定に基づくサンプルサイズの設計方法

2 つの母分散の比を

$$F(n_1, n_2; 1 - \alpha/2) \frac{V_1}{V_2} < \frac{\sigma_1^2}{\sigma_2^2} < F(n_1, n_2; \alpha/2) \frac{V_1}{V_2}$$

として区間推定するとき , 信頼上限と信頼下限の比 $\frac{F(n_1, n_2; 1 - \alpha/2)}{F(n_1, n_2; \alpha/2)}$ が δ 以下になるようにサンプルサイズ $n = n_1 = n_2$ を設計する .


表示 12.5: 2 つの母分散の比の区間推定に基づくサンプルサイズの設計方法

n	α	信頼限界比		
15	0.05	8.53		例12.4, p.191
D24:	=FINV(B24/2,A24-1,A24)/FINV(1-B24/2,A24-1,A24)			

なお , この比は $n_1 = n_2$ とするとき , n と α だけで決まる . 横軸に n を , 縦軸に母標準偏差の 信頼上限/信頼下限 の比をとったグラフを表示 12.3 に示す .

n が小さい部分は左の目盛り , 大きい部分は右の目盛りに対応する .

表示12.6: n と標準偏差の比の 上限/下限 の比の関係



12.5 2つの母平均の差の区間推定に基づくサンプルサイズ的设计方法

2つの母平均の差を

$$(\bar{x}_1 - \bar{x}_2) - t(n_1 + n_2 - 2, \alpha) \sqrt{V \left(\frac{1}{n_1} + \frac{1}{n_2} \right)} < \mu_1 - \mu_2 < (\bar{x}_1 + \bar{x}_2) + t(n_1 + n_2 - 2, \alpha) \sqrt{V \left(\frac{1}{n_1} + \frac{1}{n_2} \right)}$$

として区間推定するとき，信頼区間の幅 $2t(2n-2, \alpha) \frac{c^* \sigma \sqrt{2}}{\sqrt{n}}$ が δ 以下になるようにサンプルサイズ $n = n_1 = n_2$ を設計する．

表示12.7: 2つの母平均の差の区間推定に基づくサンプルサイズ的设计方法

n	α	σ	信頼幅			
9	0.05	1	1.97	◀▶	例12.5, p.193	9
D31:	=2*TINV(B31,2*(A31-1))*C31/SQRT(A31/2)*SQRT(2/(2*(A31-1)))*EXP(GAMMALN((2*(A31-1)+1)/2)-GAMMALN((2*(A31-1))/2))					


12.6 対応のある場合の母平均の差の区間推定に基づくサンプルサイズの設計方法

2つの母平均の差を

$$\bar{d} - t(n-1, \alpha) \sqrt{\frac{V_d}{n}} < \mu_1 - \mu_2 < \bar{d} + t(n-1, \alpha) \sqrt{\frac{V_d}{n}}$$

として区間推定するとき，信頼区間の幅 $2t(n-1, \alpha) \frac{c^* \sigma_d}{\sqrt{n}}$ が δ 以下になるようにサンプルサイズ n を設計する．

表示 12.8: 対応のある場合の母平均の差の区間推定に基づくサンプルサイズの設計方法

n	α	σ	信頼幅			
30	0.05	2	1.48		例12.6, p.195	30
E38:	=2*TINV(B39,A39-1)*C39/SQRT(A39) *SQRT(2/(A39-1))*EXP(GAMMALN((A39)/2)-GAMMALN((A39-1)/2))					

13 補遺

13.1 統計数値表

原著の最後の付表には、検定、検出力、サンプルサイズの計算に使うために統計数値表が示されている。

これらの表の値はExcelで簡単に計算することができる。

ここに取り上げられている4つの分布の表の代わりとして用いられるExcelシートを表示13.1に示す。

表示 13.1: 統計数値表

統計数値表		下の表の黒枠内に入力する.				
				入力する確率は0.50以下		
分布	x	自由度		下側確率	両側確率	上側確率
標準正規分布	2.000			0.977	0.046	0.023
	1.960				0.050	
	1.645					0.050
カイ2乗分布	20.483	10		0.975	0.050	0.025
	3.247	10		0.025		
	18.307	10				0.050
t分布	1.960	10		0.961	0.078	0.039
	2.228	10			0.050	
	1.812	10				0.050
F分布	3.456	2	5	0.886	0.228	0.114
	0.025	2	5	0.025		
	1.107	2	5			0.400

黒枠の中に入力すると、同じ行に結果が得られる。

表示13.1では小数点以下3桁まで表示しているが、桁数を増やすことが可能である。計算精度としては、有効数字が4桁以上の精度を持っている。

13.2 検出力の計算表の作成

原著の各章の第2節には、表の見出し、表頭、表側にパラメータを取り、検出力の表が作られ、それから作成されたグラフが示されている。

本テキストに示されている検出力の計算表から、同様の表とグラフを作る方法を解説する。

例として、表3.2、図3.4(p.37) を表示3.2 から作成する場合を取り上げて説明する。

表示13.2: 表3.2 の作成手順

	A	B	C	D	E	F	G	H	I	J	K	L
1	(1)				1-β			(5)	α =	0.05		
2		n	α	Δ	上側	下側	両側		Δ	n		
3		9	0.05	0.6	0.562	0.000	0.437			9	16	25
4									-1.0	0.000	0.000	0.000
5	(2)				1-β				-0.8	0.000	0.000	0.000
6		n	α	Δ	上側	下側	両側		-0.6	0.000	0.000	0.000
7		9	0.05	0.6	0.562	0.000	0.437		-0.4	0.002	0.001	0.000
8									-0.2	0.012	0.007	0.004
9	(3)	α =	0.05						-0.1	0.026	0.020	0.016
10		Δ	n						-0.05	0.036	0.033	0.029
11			9	16	25				0.00	0.050	0.050	0.050
12		0.6	0.562						0.05	0.067	0.074	0.082
13									0.1	0.089	0.107	0.126
14	(4)	α =	0.05						0.2	0.148	0.199	0.260
15		Δ	n						0.4	0.328	0.482	0.639
16			9	16	25				0.6	0.562	0.775	0.912
17		0.6	0.562						0.8	0.775	0.940	0.991
18									1.0	0.912	0.991	1.000

(1) シート「3章」からコピーする。E3のセルの内容は

=NORMSDIST(D3*SQRT(B3)-NORMSINV(1-C3))

である。

(2) 罫線を除き、字体を元に戻す。

(3) 表の枠組みを作り、 n , α , Δ を表の上と左に、上側の検出力を表中の左上のセルに移動する（コピーではないことに注意）。C12のセルの内容は

=NORMSDIST(B12*SQRT(C11)-NORMSINV(1-C9))

となり、表頭、表側から計算されるようになっていることが分かる。

(4) 検出力を計算する式に次のように \$ を挿入する。

=NORMSDIST(\$B17*SQRT(C\$16)-NORMSINV(1-\$C\$14))

これにより、コピーをしても、 α は表の上のセルに固定し、 Δ は表側列に、 n は表頭行に固定される。


(5) 表側、表頭を整え、検出力のセルを右と下にコピーする。罫線などを整形すると求める表が作成される。

表示13.2は手順を示すために、途中経過を残しているが、実際に作成するときは、(1)の上に(2)を、(3)の上に(4),(5)を作成することができる。

表示13.2のI3:L18の範囲を指定して、グラフウィザードから散布図(点を滑らかな線で結ぶ)を選択する。

得られた散布図について、整形⁵をすると表示13.3のグラフが得られる。

表示13.3: 図3.4



表の α , Δ , n を変更すると、自動的に表の内容が再計算され、グラフも更新される。


⁵ 表示13.3のグラフには次の整形が施されている。

- (1) 背景を空白にする。
- (2) 縦軸の目盛線を点線にする。
- (3) 縦軸と横軸の目盛りの範囲を修正し、「フォントの自動サイズ調整」をオフとし、「表示形式」を小数点以下1桁に指定する。
- (4) 横軸と縦軸に変数名を追加する。
- (5) 曲線の「スタイル」と「太さ」を変更する(白黒で印刷したとき凡例との対応がつくようにするため)。
- (6) 凡例の位置を移動し、グラフ本体をできるだけ大きくする。

13.3 スクロールバー

Excel シートのトップメニューから「表示」、「ツールバー」、「フォーム」を選択すると表示 13.4 の左のような ツールバー が表示される．左下から 3 つめのツールが スクロールバー である．

表示 13.4: スクロールバーの設定



「スクロールバー」をクリックして、表示したいセルをクリックすると大きな縦のバーが表示される．バーの右下のコーナーを操作すると、小さな横のバーに変形することができる．

バーを右クリックして、「コントロールの書式設定」を選択すると、表示 13.4 右の設定画面が現われる．「最小値」を 2 に設定し、「リンクするセル」に n のセルを指定する⁶．

⁶ 表示 3.3 では最小値を 1 に設定する．最大値は初期値の 100 をそのままにしてある． n は高々 10 である場合や特別に大きな n が必要の場合は適当に変更する．

変化の増分は整数だけが許される．たとえば、 Δ を 0.1 きざみで変化させて、検出力の変化を見たいというような場合には、スクロールバーで制御されるセル（整数）を別に設定し、その値を 10 で割って Δ を計算する．

13.4 ユーザー定義関数

NTDIST, NFDIST, NCDIST は 非心 t 分布, 非心 F 分布, 非心カイ 2 乗分布 の累積分布関数 (下側確率) を求めるためのユーザー定義関数である。

これらの関数の引数は, t 分布, F 分布, カイ 2 乗分布の関数 TDIST, FDIST, CHIDIST の引数の最後に非心パラメータを追加したものである。

TDIST 関数は, t は正の値のみが許され, 最後に片側, 両側を指定するパラメータによって上側確率 または上側確率の 2 倍が得られる。

FDIST, CHIDIST の両関数は, 上側確率が得られる。

それに対して, 3 つの非心分布の関数はすべて 下側確率 が求められるようになっている (NORMSDIST 関数と同じ)。また, NTDIST 関数は, t の値が正負いずれでも構わない。

NTDIST 関数は, 非心パラメータのあとに, 1 を追加すると, 確率密度が求められるというオプションが準備されている。

これらのプログラムは VBA で書かれており, Excel でマクロの内容を見ることができる。

これらは, 下記の本に記載されている FORTRAN プログラムを VBA に書き換えたものである⁷。

「統計数値表」, 日本規格協会 (1972)

非心 t 分布の上側確率	p.246
非心 F 分布の上側確率	p.248
非心 χ^2 分布の上側確率	p.250
T 関数 (非心 χ^2 分布 で利用)	p.245

⁷ 非心 F 分布の FORTRAN プログラムには誤りがあり, 分子の自由度が 1, 2 のときの結果が正しくない。

行番号 10 を 1 行上に移すと正しい結果が得られる。

13.5 JMP による検出力の計算

統計解析プログラム JMP には、 t 分布、 2 乗分布、 F 分布で、オプションとして非心パラメータを指定することができる。これらの関数を利用すると、Excel で作成した検出力の計算表と同様のものを比較的簡単に作成することができる。

Excel はセルに式が入力されるのに対して、JMP は列に式が入力される。

表示13.5 には、表示3.2、表示5.2 に代わる計算表を示す。

表示13.5: 1つの母平均の検定の検出力

	n	α	Δ	$1-\beta_3$	$1-\beta_5$	参照
1	9	0.05	0.6	0.562	0.500	例3.5, 5.3
2	9	0.025	0.6	0.436	0.354	例3.3, 5.1
3	44	0.05	0.5	0.953	0.947	例3.10
4	25	0.05	0.5	0.804	0.783	例3.11
5	11	0.025	1	0.913	0.848	例3.9
6	45	0.05	0.5	0.956	0.951	例5.8
7	27	0.05	0.5	0.830	0.812	例5.9
8	13	0.025	1	0.950	0.911	例5.7

表示3.2、表示5.2 では、上側、下側、両側の3つの検出力が求められているが、表示13.5 では上側の検出力だけを示す。

下側の検出力を求めたいときは、 Δ の符号を負にする。また、両側の検出力は α を半分にして入力する。

$1-\beta_3$ 、 $1-\beta_5$ には、§3 (σ 既知)、§5 (σ 未知) の場合の検出力である。それぞれの列には次の計算式が入力されている。

$1-\beta_3$: $1-\text{NormDistribution}(\text{NormQuantile}(1-\alpha)-\Delta * \text{Root}(n))$

$1-\beta_5$: $1-\text{tDistribution}(\text{tQuantile}(1-\alpha, n-1), n-1, \Delta * \text{Root}(n))$


なお、 tDistribution 、 tQuantile は Excel の TDIST 、 TINV 関数と異なり、正規分布と同様、下側確率が用いられている。従って、上の2つの計算式はほぼ同じ形をしている。

tDistribution の3番目のパラメータ $\Delta \sqrt{n}$ は非心パラメータである。

JMP にはスクロールバーがないので、 n を試行錯誤で変化させなければならない。

表示13.6 は 表示7.1 の計算表に代わるものである .

表示 13.6: 2 つの母平均の差の検定の検出力



	n 1	n 2	α	Δ	$1 - \beta$	参考
1	10	8	0.05	0.6	0.332	例7.3
2	10	8	0.025	0.6	0.221	例7.1
3	11	11	0.05	1.5	0.960	例7.8
4	10	10	0.05	1.2	0.825	例7.9
5	22	22	0.025	1	0.900	例7.7

計算式は

$1 - \text{tDistribution}(\text{tQuantile}(1 - \alpha, n1 + n2 - 2), n1 + n2 - 2, \Delta * \text{Root}(n1 * n2 / (n1 + n2)))$

である .

他の章の計算表も JMP で可能であるが , 省略する .